Maple-assisted proof of empirical formula for A267242

Robert Israel

Sep 8 2019

Consider a "state" of the system to be a 1×9 binary array x, where $x_1 .. x_5$ represent a row, and for j from 6 to 9, $x_j = 1$ if this row and the previous ones have already determined that column $j-4$ is lexicographically greater than column $j-5$. In particular if $x_1 \neq x_2$ we must have $x_6 = 1$ and similarly for the others.

We enumerate the 162 possible states:

```
> S[2]:= [[0,0,0],[0,0,1],[0,1,1],[1,0,1],[1,1,0],[1,1,1]]:
  for i from 3 to 5 do
    S[i]:= map(proc(t) [op(t[1..i-1]),t[i-1],op(t[i..-1]),0], [op(t
  [1..i-1]),t[i-1],op(t[i..-1]),1],
       [op(t[1..i-1]),1-t[i-1],op(t[i..-1]),1] end proc, S[i-1])
  od:
  states:= S[5]:
  nops(states);
```

$$162 \quad (1)$$

Although it is not part of the $n \times 5$ array, we may imagine that we start in state [0,0,0,0,0,0,0,0,0]. Let T be the 162×162 transition matrix where $T_{ij} = 1$ if state j can be followed by state i.

```
> T:= Matrix(162,162,proc(i,j) local k;
    if add(states[j,k]-states[i,k],k=1..5) > 0 then return 0 fi;
    for k from 6 to 9 do if states[j,k]>states[i,k] then return 0
  fi od;
    for k from 1 to 4 do if states[i,k]>=states[i,k+1] and states
  [j,k+5]<>states[i,k+5] then return 0 fi od;
  1
  end proc):
```

Then we should have $a_n = u^T T^n e$ where $u = (1, ..., 1)^T$ and $e = (1, 0, ..., 0)^T$. To check, we compute the first few terms of the sequence. .

```
> E:= Vector(162): E[1]:= 1:
  U[0]:= Vector[row](162,1):
  for k from 1 to 25 do U[k]:= U[k-1].T od:
  seq(U[j] . E, j=1..25);
```

6, 34, 232, 1986, 20040, 220235, 2499080, 28501471, 323067002, 3626695952, 40306404192, 443852375808, 4848323701804, 52590398731297, 567018802063680, 6081537709403509, 64929807220896558, 690446673537426382, 7315972510023985656, 77274873614152163086, 813900638760063735376, 8550527339833707689319, 89620491846900286358232, 937357144562779343633051, 9785093901830068685014370 **(2)**

Now the empirical formula is

```
> Emp:=  a(n) = 52*a(n-1) -1196*a(n-2) +16140*a(n-3) -142918*a(n-4)
  +879116*a(n-5) -3875668*a(n-6) +12442580*a(n-7) -29232481*a(n-8)
  +50015232*a(n-9) -61355336*a(n-10) +52355680*a(n-11) -29405200*a
  (n-12) +9744000*a(n-13) -1440000*a(n-14):
```

This corresponds to $u^T P(T) T^n e = 0$ where $P(x)$ is the following polynomial of degree 15:

```
> P:= x^14 - add(coeff(rhs(Emp),a(n-i))*x^(14-i),i=1..14);
```

$$P := x^{14} - 52\,x^{13} + 1196\,x^{12} - 16140\,x^{11} + 142918\,x^{10} - 879116\,x^9 + 3875668\,x^8 - 12442580\,x^7 + 29232481\,x^6 - 50015232\,x^5 + 61355336\,x^4 - 52355680\,x^3 + 29405200\,x^2 - 9744000\,x + 1440000 \quad \textbf{(3)}$$

It turns out that $u^T P(T) = 0$. The verification of this completes the proof.

```
> UP:= add(coeff(P,x,j)*U[j],j=0..14):
  UP . UP^%T;
```

$$0 \quad \textbf{(4)}$$